```
> restart;with(DEtools):with(linalg):with(plots):
```

How to carry out an undetermined coefficients calculation with Maple assistance.

This script helps do all the necessary computations as if you were doing them by hand.

Maple simply keeps track of all operations so there are no errors. This format is acceptable

for the homework

(Text, problem 4.8.37).

Consider the ODE (linear, homogeneous, constant coefficients):

```
> de1 := D(D(y))(x) - 4*D(y)(x) + 5*y(x) =
> exp(5*x)+x*sin(3*x)-cos(3*x);
```

$$de1 := (D^{(2)})(y)(x) - 4\,D(y)(x) + 5\,y(x) = e^{(5\,x)} + x\sin(3\,x) - \cos(3\,x)$$

```
> char:= r^2-4*r+5=0;
```

$$char := r^2 - 4\,r + 5 = 0$$

The roots of the characteristic equation do not match those for the rhs: no resonance.

```
> solve(%,r);
```

$$2 + I,\ 2 - I$$

```
> yg:=a1*exp(5*x)+(a2+a3*x)*cos(3*x)+(a4+a5*x)*sin(3*x);
```

$$yg := a1\,e^{(5\,x)} + (a2 + a3\,x)\cos(3\,x) + (a4 + a5\,x)\sin(3\,x)$$

```
> y0:=5*yg;
```

$$y0 := 5\,a1\,e^{(5\,x)} + 5\,(a2 + a3\,x)\cos(3\,x) + 5\,(a4 + a5\,x)\sin(3\,x)$$

```
> yp:=diff(yg,x);
```

$$yp := 5\,a1\,e^{(5\,x)} + \cos(3\,x)\,a3 - 3\,(a2 + a3\,x)\sin(3\,x) + \sin(3\,x)\,a5 + 3\,(a4 + a5\,x)\cos(3\,x)$$

```
> y1:=-4*yp;
```

$$y1 := -20\,a1\,e^{(5\,x)} - 4\cos(3\,x)\,a3 + 12\,(a2 + a3\,x)\sin(3\,x) - 4\sin(3\,x)\,a5 - 12\,(a4 + a5\,x)\cos(3\,x)$$

```
> y2:=diff(yp,x);
```

$$y2 := 25\,a1\,e^{(5\,x)} - 6\sin(3\,x)\,a3 - 9\,(a2 + a3\,x)\cos(3\,x) + 6\cos(3\,x)\,a5 - 9\,(a4 + a5\,x)\sin(3\,x)$$

```
> Y:= simplify(y0+y1+y2-exp(5*x)-x*sin(3*x)+cos(3*x));
```

$$Y := 10\,a1\,e^{(5\,x)} - 4\cos(3\,x)\,a2 - 4\cos(3\,x)\,a3\,x - 4\sin(3\,x)\,a4 - 4\sin(3\,x)\,a5\,x - 4\cos(3\,x)\,a3 + 12\sin(3\,x)\,a2 + 12\sin(3\,x)\,a3\,x - 4\sin(3\,x)\,a5 - 12\cos(3\,x)\,a4 - 12\cos(3\,x)\,a5\,x - 6\sin(3\,x)\,a3 + 6\cos(3\,x)\,a5 - e^{(5\,x)} - x\sin(3\,x) + \cos(3\,x)$$

```
> Y1:=10*a1-1=0;
```

$$Y1 := 10\,a1 - 1 = 0$$

```
> Y2:=-4*a2-4*a3-12*a4+6*a5+1=0;
```

$$Y2 := -4\,a2 - 4\,a3 - 12\,a4 + 6\,a5 + 1 = 0$$

```
> Y3:=-4*a4+12*a2-4*a5-6*a3=0;
```

$$Y3 := -4\,a4 + 12\,a2 - 4\,a5 - 6\,a3 = 0$$

```
> Y4:=-4*a3-12*a5=0;
```

$$Y4 := -4\,a3 - 12\,a5 = 0$$

```
> Y5:=-4*a5+12*a3-1=0;
```

$$Y5 := -4\,a5 + 12\,a3 - 1 = 0$$

```
> solve({Y1,Y2,Y3,Y4,Y5},{a1,a2,a3,a4,a5});
```

$$\{a1 = \frac{1}{10},\ a5 = \frac{-1}{40},\ a4 = \frac{13}{400},\ a3 = \frac{3}{40},\ a2 = \frac{1}{25}\}$$

```
> init_con := y(0) = 0, D(y)(0) = 0;
```

$$init\_con := \mathrm{y}(0) = 0,\ \mathrm{D}(y)(0) = 0$$

We now compute the solution using DSOLVE (set C1=C2=0 for particular solution):

```
> gsolution := dsolve({de1},y(x));
```

$$gsolution := \{\mathrm{y}(x) = e^{(2\,x)}\sin(x)\,\_C2 + e^{(2\,x)}\cos(x)\,\_C1 + \frac{1}{400}(16 + 30\,x)\cos(3\,x) + \frac{1}{10}e^{(5\,x)} + \frac{13}{400}\sin(3\,x) - \frac{1}{40}x\sin(3\,x)\}$$

```
> IVP := {de1,init_con};soln := dsolve(IVP, y(x));expr :=
> subs(soln,y(x)):
```

$$IVP := \{(\mathrm{D}^{(2)})(y)(x) - 4\,\mathrm{D}(y)(x) + 5\,\mathrm{y}(x) = e^{(5\,x)} + x\sin(3\,x) - \cos(3\,x),\ \mathrm{D}(y)(0) = 0,\ \mathrm{y}(0) = 0\}$$

$$soln := \mathrm{y}(x) = -\frac{157}{400}e^{(2\,x)}\sin(x) - \frac{7}{50}e^{(2\,x)}\cos(x) + \frac{1}{400}(16 + 30\,x)\cos(3\,x) + \frac{1}{10}e^{(5\,x)} + \frac{13}{400}\sin(3\,x) - \frac{1}{40}x\sin(3\,x)$$

```
> plot(expr,x=-15..0.6, axes=BOXED,title="solution to a 2nd order
> IVP");
```


solution to a 2nd order IVP
1
0.5
0
−0.5
−1
−14
−12
−10
−8
−6
−4
−2
0
x
Legend
Curve 1